
    OpenIDConnect

Client library for consuming and working with OpenID Connect Providers
OpenIDConnect is built and maintained by DockYard, contact us for expert Elixir and Phoenix consulting.
Installation
Available in Hex, the package can be installed as:
Add openid_connect to your list of dependencies in mix.exs:
def deps do
  [{:openid_connect, "~> 1.0.1"}]
end
Getting Started
Configuration
This library works with any OpenID Connect provider through a standard configuration map. You'll need to create a configuration map for each provider you want to work with:
google_config = %{
  discovery_document_uri: "https://accounts.google.com/.well-known/openid-configuration",
  client_id: "CLIENT_ID",
  client_secret: "CLIENT_SECRET",
  redirect_uri: "https://example.com/session",
  response_type: "code",
  scope: "openid email profile"
}
The configuration requires:
	discovery_document_uri: URL to the OpenID Connect Discovery Document
	client_id: OAuth 2.0 Client Identifier provided by your identity provider
	client_secret: OAuth 2.0 Client Secret provided by your identity provider
	response_type: OAuth 2.0 Response Type (typically "code" for authorization code flow)
	scope: Space-separated list of scopes (must include "openid")

Additional optional configuration:
	leeway: Number of seconds to allow for clock skew when verifying token expiration (default: 30)

Most major OAuth2 providers support OpenID Connect, including:
	Google
	Microsoft Azure AD
	Auth0
	Okta
	AWS Cognito
	Keycloak
	OneLogin
	Many others

For a full list, see major OpenID Connect providers.
Usage
Implementing OpenID Connect authentication in your app involves these steps:
	Generate an authorization URI and redirect the user to it
	Handle the callback from the provider after successful authentication
	Exchange the authorization code for tokens
	Verify the ID token
	Use the claims to establish a user session

1. Generate the Authorization URI
Create a URL that will redirect the user to the provider's login page:
{:ok, uri} = OpenIDConnect.authorization_uri(
  google_config,
  "https://example.com/auth/callback",
  %{state: state_token}
)
You should:
	Generate a secure state token for CSRF protection
	Store this state token in your session
	Include the state token in the authorization URI

2. Handle the Callback from the Provider
After the user logs in, the provider redirects back to your redirect_uri with an authorization code and the state token you provided:
# In your callback controller/handler
def callback(conn, params) do
  # First verify that the state parameter matches the one we stored
  stored_state = get_session(conn, :state_token)
  received_state = params["state"]
  
  if Plug.Crypto.secure_compare(stored_state, received_state) do
    # State is valid, proceed with code exchange
    handle_valid_callback(conn, params)
  else
    # State token doesn't match, reject the request
    conn
    |> put_status(401)
    |> render("error.html", error: "Invalid state parameter")
  end
end
3. Exchange the Authorization Code for Tokens
Exchange the code for ID and access tokens:
{:ok, tokens} = OpenIDConnect.fetch_tokens(google_config, %{
  code: params["code"],
  redirect_uri: "https://example.com/auth/callback"  # Must match original redirect_uri
})

# tokens will contain:
# %{
#   "access_token" => "ya29.a0AfB_...",
#   "id_token" => "eyJhbGciOiJSUzI1...",
#   "expires_in" => 3599,
#   "token_type" => "Bearer",
#   "refresh_token" => "1//03s..." (if requested)
# }
4. Verify the ID Token
The ID token is a JWT containing claims about the user. Always verify it before trusting its contents:
{:ok, claims} = OpenIDConnect.verify(google_config, tokens["id_token"])

# claims will contain user information like:
# %{
#   "sub" => "10865933565....",  # Unique user identifier
#   "email" => "user@example.com",
#   "email_verified" => true,
#   "name" => "User Name",
#   "picture" => "https://...",
#   "given_name" => "User",
#   "family_name" => "Name",
#   "locale" => "en",
#   "iat" => 1585662674,  # Issued at timestamp
#   "exp" => 1585666274,  # Expiration timestamp
#   "aud" => "YOUR_CLIENT_ID",
#   "iss" => "https://accounts.google.com"
# }
5. Use the Claims to Establish a User Session
Use the verified claims to identify your user and establish a session:
# The "sub" claim is the unique identifier for the user
user_id = claims["sub"]

# Find or create a user in your system
user = find_or_create_user(user_id, claims)

# Establish a session for this user
conn
|> put_session(:user_id, user.id)
|> redirect(to: "/dashboard")
Optional: Fetch Additional User Information
Some providers allow you to fetch additional user information using the access token:
{:ok, userinfo} = OpenIDConnect.fetch_userinfo(google_config, tokens["access_token"])
Ending a User Session
Some providers support log out via an end session endpoint:
{:ok, logout_uri} = OpenIDConnect.end_session_uri(
  google_config,
  %{id_token_hint: id_token, post_logout_redirect_uri: "https://example.com/logout/callback"}
)

# Redirect the user to the logout_uri
redirect(conn, external: logout_uri)
Key Security Considerations:
	Always validate the state token
	Store secrets (client ID, client secret) in environment variables
	Verify the ID token before trusting its contents
	Use HTTPS for all redirect URIs
	Consider token storage carefully - sessions are usually appropriate
	Clear tokens when logging out

Authors
	Brian Cardarella

We are very thankful for the many contributors
Versioning
This library follows Semantic Versioning
Looking for help with your Elixir project?
At DockYard we are ready to help you build your next Elixir project. We have a unique expertise in Elixir and Phoenix development that is unmatched. Get in touch!
At DockYard we love Elixir! You can read our Elixir blog posts
or come visit us at The Boston Elixir Meetup that we organize.
Want to help?
Please do! We are always looking to improve this library. Please see our
Contribution Guidelines
on how to properly submit issues and pull requests.
Sponsors
In addition to our contributors, this library receives support and maintenance from TV Labs

    Complete Phoenix Example

Here's a complete example of implementing OpenID Connect authentication in a Phoenix application:
# Configuration in config/config.exs
config :my_app, :google_oidc_config, %{
  discovery_document_uri: "https://accounts.google.com/.well-known/openid-configuration",
  client_id: System.get_env("GOOGLE_CLIENT_ID"),
  client_secret: System.get_env("GOOGLE_CLIENT_SECRET"),
  response_type: "code",
  scope: "openid email profile"
}

# Routes in lib/my_app_web/router.ex
scope "/auth", MyAppWeb do
  pipe_through [:browser, :redirect_if_authenticated]
  
  get "/google", AuthController, :request
  get "/google/callback", AuthController, :callback
  get "/logout", AuthController, :logout
end

# Controller in lib/my_app_web/controllers/auth_controller.ex
defmodule MyAppWeb.AuthController do
  use MyAppWeb, :controller
  
  @doc """
  Initiates the OpenID Connect authentication flow
  """
  def request(conn, _params) do
    google_config = Application.fetch_env!(:my_app, :google_oidc_config)
    
    # Generate a secure state token to prevent CSRF
    state_token = Base.encode64(:crypto.strong_rand_bytes(32), padding: false)
    
    # The redirect_uri must match what your provider has configured
    redirect_uri = Routes.auth_url(conn, :callback)
    
    {:ok, auth_uri} = OpenIDConnect.authorization_uri(
      google_config,
      redirect_uri,
      %{state: state_token}
    )
    
    conn
    |> put_session(:oidc_state_token, state_token)
    |> redirect(external: auth_uri)
  end
  
  @doc """
  Handles the callback from the OpenID provider after successful authentication
  """
  def callback(conn, params) do
    google_config = Application.fetch_env!(:my_app, :google_oidc_config)
    
    # Verify the state parameter to prevent CSRF attacks
    with true <- valid_state_token?(conn, params["state"]),
         {:ok, tokens} <- OpenIDConnect.fetch_tokens(
           google_config,
           %{
             code: params["code"],
             redirect_uri: Routes.auth_url(conn, :callback)
           }
         ),
         {:ok, claims} <- OpenIDConnect.verify(google_config, tokens["id_token"]) do
      
      # Store tokens securely, typically in the session or a secure cookie
      # Storing in the session should be fine for most cases
      conn = put_session(conn, :access_token, tokens["access_token"])
      
      # Find or create the user based on the claims
      user = 
        case Accounts.get_user_by_provider_id("google", claims["sub"]) do
          nil -> 
            {:ok, user} = Accounts.create_user(%{
              email: claims["email"],
              name: claims["name"],
              provider: "google",
              provider_id: claims["sub"]
            })
            user
          user -> 
            user
        end
      
      # Establish the user's session
      conn
      |> put_session(:current_user_id, user.id)
      |> configure_session(renew: true)
      |> put_flash(:info, "Successfully signed in!")
      |> redirect(to: Routes.dashboard_path(conn, :index))
    else
      false -> 
        # Invalid state token, potential CSRF attack
        conn
        |> put_flash(:error, "Authentication failed: Invalid state parameter")
        |> redirect(to: Routes.page_path(conn, :index))
      
      {:error, reason} ->
        # Log the error for debugging
        require Logger
        Logger.error("Authentication error: #{inspect(reason)}")
        
        conn
        |> put_flash(:error, "Authentication failed")
        |> redirect(to: Routes.page_path(conn, :index))
    end
  end
  
  @doc """
  Logs the user out by clearing the session
  """
  def logout(conn, _params) do
    google_config = Application.fetch_env!(:my_app, :google_oidc_config)
    id_token = get_session(conn, :id_token)
    
    # Clear our local session
    conn = 
      conn
      |> configure_session(drop: true)
      |> put_flash(:info, "Successfully signed out!")
    
    # If we have an ID token, try to use the provider's logout endpoint
    if id_token do
      case OpenIDConnect.end_session_uri(
        google_config,
        %{
          id_token_hint: id_token,
          post_logout_redirect_uri: Routes.page_url(conn, :index)
        }
      ) do
        {:ok, logout_uri} ->
          redirect(conn, external: logout_uri)
        _ ->
          # Fall back to local logout if provider doesn't support end_session
          redirect(conn, to: Routes.page_path(conn, :index))
      end
    else
      redirect(conn, to: Routes.page_path(conn, :index))
    end
  end
  
  # Verify the state token to prevent CSRF attacks
  defp valid_state_token?(conn, token) do
    stored_token = get_session(conn, :oidc_state_token)
    
    # Clear the token to prevent replay attacks
    conn = delete_session(conn, :oidc_state_token)
    
    case stored_token do
      nil -> false
      ^token -> true
      _ -> false
    end
  end
end

    CHANGELOG

v1.0.1
Added
	Add option ignore_claims to verify/3 for optionally ignoring claim verification

v1.0.0
Complete rewrite of the library by @AndrewDryga
Added
	Add end_session_uri/2 and fetch_userinfo/2
	Adds OpenID claim validation

Removed
	GenServer bottleneck
	Atom requirement for provider name
	Application config from the library

Changed
	Rewrote tests to better cover production code
	Use Finch/Mint as the HTTP client instead of HTTPoison

Please see the documentation for migrating from prior versions.
v0.2.2
	Allow missing claims_supported in discovery document
	Allow overriding document params

v0.2.1
	Relaxed jason version requirement

v0.2.0
	BREAKING CHANGE - Multiple response types supported. See: PR #17 for details

v0.1.2
	Optional params for authorization_uri

v0.1.1
	Remove Poison (legacy dep)
	Force JOSE to use Jason in test suite

v0.1.0
	Initial public release

OpenID Connect client library for Elixir.
This library provides a complete implementation of the OpenID Connect 
authentication protocol, which is built on top of OAuth 2.0. It handles the complete authentication 
flow including:
	Generating authorization URIs for redirecting users to identity providers
	Fetching tokens from the token endpoint
	Verifying JWT ID tokens using provider JWKs (JSON Web Keys)
	Fetching user information from the userinfo endpoint
	Supporting provider logout via end session endpoints

Supported Authentication Flows
The library supports the standard OpenID Connect authentication flows:
	Authorization Code Flow (most common for web applications)
	Implicit Flow (less secure, used for JavaScript applications)
	Hybrid Flow (combines aspects of both)

Supported Identity Providers
This library works with any OpenID Connect compliant provider, including:
	Google
	Microsoft Azure AD
	Auth0
	Okta
	Amazon Cognito
	Keycloak
	OneLogin
	HashiCorp Vault
	And many others

Basic Usage
# Step 1: Configure the provider
google_config = %{
  discovery_document_uri: "https://accounts.google.com/.well-known/openid-configuration",
  client_id: "YOUR_CLIENT_ID",
  client_secret: "YOUR_CLIENT_SECRET",
  response_type: "code",
  scope: "openid email profile"
}

# Step 2: Generate the authorization URI (redirect the user to this URI)
{:ok, uri} = OpenIDConnect.authorization_uri(
  google_config,
  "https://example.com/auth/callback",
  %{state: state_token}
)

# Step 3: Exchange the authorization code for tokens
{:ok, tokens} = OpenIDConnect.fetch_tokens(
  google_config,
  %{
    code: auth_code,
    redirect_uri: "https://example.com/auth/callback"
  }
)

# Step 4: Verify the ID token
{:ok, claims} = OpenIDConnect.verify(google_config, tokens["id_token"])

# Optional: Fetch additional user information
{:ok, userinfo} = OpenIDConnect.fetch_userinfo(google_config, tokens["access_token"])
See the README for more detailed examples and configuration options.

        Fetches authentication tokens from the provider using the token endpoint.

          @type client_id() :: String.t()


      


OAuth 2.0 Client Identifier valid at the Authorization Server.

  



  
    
      
    
    
      client_secret()



        
          
        

    

  


  

      

          @type client_secret() :: String.t()


      


OAuth 2.0 Client Secret valid at the Authorization Server.

          @type config() :: %{
  :discovery_document_uri => discovery_document_uri(),
  :client_id => client_id(),
  :client_secret => client_secret(),
  :response_type => response_type(),
  :scope => scope(),
  optional(:leeway) => non_neg_integer()
}


      


The configuration of a OpenID provider.

          @type discovery_document_uri() :: String.t()


      


URL to a OpenID Discovery Document endpoint.

          @type jwt() :: String.t()


      


JSON Web Token
See: https://jwt.io/introduction/

          @type redirect_uri() :: String.t()


      


Redirection URI to which the response will be sent.
This URI MUST exactly match one of the Redirection URI values for the Client pre-registered at the OpenID Provider,
with the matching performed as described in Section 6.2.1 of [RFC3986] (Simple String Comparison).
When using this flow, the Redirection URI SHOULD use the https scheme; however, it MAY use the http scheme,
provided that the Client Type is confidential, as defined in Section 2.1 of OAuth 2.0,
and provided the OP allows the use of http Redirection URIs in this case. The Redirection URI MAY use an alternate scheme,
such as one that is intended to identify a callback into a native application.

          @type response_type() :: [String.t()] | String.t()


      


OAuth 2.0 Response Type value that determines the authorization processing flow to be used,
including what parameters are returned from the endpoints used.

          @type scope() :: [String.t()] | String.t()


      


OAuth 2.0 Scope Values that the Client is declaring that it will restrict itself to using.

          @type verify_opts() :: [{:ignore_claims, [String.t()]}]


      


Options for OpenIDConnect.verify/3:
	:ignore_claims - A list of claims to not be verified.
Available claim options to use: "aud".

          @spec authorization_uri(
  config(),
  redirect_uri :: redirect_uri(),
  params :: %{optional(atom()) => term()}
) :: {:ok, uri :: String.t()} | {:error, term()}


      


Builds an authorization URI for redirecting users to the identity provider.
This function creates a URL that starts the OpenID Connect authentication flow
by redirecting the user to the provider's authorization endpoint. The endpoint URL
is automatically retrieved from the provider's discovery document.
Parameters
	config - The provider configuration map
	redirect_uri - URI to redirect to after authentication (must match one registered with the provider)
	params - Optional map of additional query parameters to include in the authorization URI

Common Additional Parameters
	state - Strongly recommended - Random token to prevent CSRF attacks
	prompt - Controls the login experience, common values include:	"none" - No interactive prompt, fails if user authentication is required
	"login" - Forces the user to enter their credentials even if already logged in
	"consent" - Forces the consent screen to be displayed even if previously consented
	"select_account" - Prompts the user to select an account


	login_hint - Email address or sub identifier to pre-fill the login form
	max_age - Maximum elapsed time in seconds since last authentication
	ui_locales - Preferred languages for the UI, space-separated list of BCP47 tags
	acr_values - Authentication Context Class Reference values

Provider-Specific Parameters
Some providers support additional custom parameters:
	Google: hd (hosted domain) to restrict to specific Google Workspace domains
	Azure AD: domain_hint to skip the home realm discovery page
	Many others based on the provider

Returns
	{:ok, uri} - On success, returns the authorization URI string
	{:error, reason} - On failure, returns an error tuple with details

Example: Basic Usage
{:ok, uri} = OpenIDConnect.authorization_uri(
  google_config,
  "https://example.com/auth/callback",
  %{state: state_token}
)
Example: With Additional Parameters
{:ok, uri} = OpenIDConnect.authorization_uri(
  google_config,
  "https://example.com/auth/callback",
  %{
    state: state_token,
    prompt: "login",
    login_hint: "user@example.com",
    hd: "example.com"  # Google-specific parameter
  }
)
Security: State Parameter
The state parameter is critical for preventing cross-site request forgery attacks.
Always include a state parameter with a secure random value and validate it when
the user is redirected back to your application.
Creating a State Token
state_token = Base.encode64(:crypto.strong_rand_bytes(32), padding: false)

# Store the token in your session
conn = Plug.Conn.put_session(conn, :oidc_state_token, state_token)

# Include it in the authorization URI
{:ok, uri} = OpenIDConnect.authorization_uri(
  google_config,
  redirect_uri,
  %{state: state_token}
)
Validating the State Token
When the provider redirects back to your application, verify the state parameter:
stored_token = Plug.Conn.get_session(conn, :oidc_state_token)
received_token = params["state"]

if stored_token && Plug.Crypto.secure_compare(stored_token, received_token) do
  # State token is valid, proceed with token exchange
else
  # Invalid state token, reject the request
end
For more details on state tokens, see Google's Documentation.

          @spec end_session_uri(config(), params :: %{optional(atom()) => term()}) ::
  {:ok, uri :: String.t()} | {:error, term()}


      


Builds a URI for ending the user's session with the identity provider.
This function creates a URL for OpenID Connect RP-Initiated Logout, allowing
your application to sign the user out of the identity provider's session.
The endpoint URL is automatically retrieved from the provider's discovery document.
Parameters
	config - The provider configuration map
	params - Optional map of additional query parameters for the end session URI

Common Additional Parameters
	id_token_hint - The ID token previously issued to the user (strongly recommended)
	post_logout_redirect_uri - URI to redirect to after logout (must be registered with provider)
	state - Opaque value for maintaining state between logout request and callback

Provider Requirements
Different providers have different requirements for logout:
	Azure AD: Requires post_logout_redirect_uri and may accept id_token_hint
	Auth0: Requires client_id and returnTo (instead of post_logout_redirect_uri)
	Cognito: Requires client_id and logout_uri (their name for redirect URI)
	Okta: Requires id_token_hint and accepts post_logout_redirect_uri
	Keycloak: Accepts id_token_hint and post_logout_redirect_uri

Always consult your provider's documentation for specific requirements.
Returns
	{:ok, uri} - On success, returns the end session URI string
	{:error, :endpoint_not_set} - If the provider doesn't support RP-Initiated Logout
	{:error, reason} - On other failures, returns an error tuple with details

Provider Support Note
Not all providers support RP-Initiated Logout. If the provider's discovery document
doesn't include an end_session_endpoint, this function will return
{:error, :endpoint_not_set}. In such cases, you'll need to implement session
termination in your application only.
Example: Azure AD Logout
{:ok, logout_uri} = OpenIDConnect.end_session_uri(
  azure_config,
  %{
    post_logout_redirect_uri: "https://example.com/logged-out"
  }
)

# Redirect the user to the logout_uri
redirect(conn, external: logout_uri)
Example: With ID Token Hint
{:ok, logout_uri} = OpenIDConnect.end_session_uri(
  okta_config,
  %{
    id_token_hint: id_token,
    post_logout_redirect_uri: "https://example.com/logged-out"
  }
)
For more details, see the OpenID Connect RP-Initiated Logout Specification.

          @spec fetch_tokens(config(), params :: %{optional(atom()) => term()}) ::
  {:ok, response :: map()} | {:error, term()}


      


Fetches authentication tokens from the provider using the token endpoint.
This function exchanges the authorization code or refresh token for access and ID tokens
from the identity provider's token endpoint. The endpoint URL is automatically retrieved
from the provider's discovery document.
Parameters
	config - The provider configuration map
	params - A map of parameters to send to the token endpoint

Common Parameters
The params map should include different fields depending on the grant type:
Authorization Code Grant (most common)
%{
  code: "AUTHORIZATION_CODE_FROM_CALLBACK",
  redirect_uri: "https://example.com/callback", # Must match the original redirect_uri
  grant_type: "authorization_code" # Optional, defaults to "authorization_code"
}
Refresh Token Grant
%{
  refresh_token: "REFRESH_TOKEN",
  grant_type: "refresh_token"
}
Returns
	{:ok, tokens} - On successful token exchange, returns a map containing at minimum:
	"access_token" - The OAuth 2.0 access token
	"token_type" - The token type (typically "Bearer")
	"expires_in" - Token lifetime in seconds
	"id_token" - The OpenID Connect ID token (JWT)
	"refresh_token" - Optional refresh token for obtaining new access tokens


	{:error, reason} - On failure, returns an error tuple with reason


Example
{:ok, tokens} = OpenIDConnect.fetch_tokens(
  google_config,
  %{
    code: params["code"],
    redirect_uri: "https://example.com/auth/callback" 
  }
)

# Access the tokens
access_token = tokens["access_token"]
id_token = tokens["id_token"]
refresh_token = tokens["refresh_token"]
For more details, see the OpenID Connect spec.

          @spec fetch_userinfo(config(), jwt()) :: {:ok, response :: map()} | {:error, term()}


      


Fetches information about the authenticated user from the userinfo endpoint.
This function retrieves additional claims about the end-user from the provider's
userinfo endpoint using the access token. This can be useful when you need user
attributes that weren't included in the ID token claims.
Parameters
	config - The provider configuration map
	access_token - The OAuth 2.0 access token obtained from fetch_tokens/2

Returns
	{:ok, userinfo} - On success, returns a map of user information
	{:error, reason} - On failure, returns an error tuple with details

Userinfo Claims
The returned claims will depend on the scopes requested during authorization,
but typically include:
	"sub" - Subject identifier (unique user ID)
	"name" - User's full name (if "profile" scope)
	"given_name" - First name (if "profile" scope)
	"family_name" - Last name (if "profile" scope)
	"email" - Email address (if "email" scope)
	"email_verified" - Boolean indicating if email is verified (if "email" scope)
	"picture" - URL to profile picture (if "profile" scope)

Example
{:ok, userinfo} = OpenIDConnect.fetch_userinfo(google_config, access_token)

# Access user information
user_id = userinfo["sub"]
email = userinfo["email"]
name = userinfo["name"]
Security Note
The userinfo endpoint requires a valid access token. The token must have
appropriate scopes (typically "openid" and others like "profile" or "email").
For more details, see the OpenID Connect UserInfo Endpoint.

          @spec verify(config(), jwt :: String.t(), verify_opts()) ::
  {:ok, claims :: map()} | {:error, term()}


      


Verifies the validity and authenticity of an OpenID Connect ID token.
This security-critical function verifies that:
	The token's signature is valid and was signed by the provider's key
	The token has not expired (checking the "exp" claim)
	The token is intended for your application (checking the "aud" claim)

Parameters
	config - The provider configuration map
	jwt - The ID token string (a JSON Web Token) from the tokens response

Options
	:ignore_claims - A list of claims to not be verified.
Available claim options to use: "aud".

Returns
	{:ok, claims} - On successful verification, returns the decoded claims from the token
	{:error, reason} - On verification failure, returns an error tuple with details

Verification Process
	The signature is verified using the provider's JSON Web Keys (JWKs), which are fetched
from the provider's JWKs endpoint listed in the discovery document.
	The "exp" (expiration) claim is checked to ensure the token has not expired.
A configurable leeway (default: 30 seconds) is allowed to account for clock skew.
	The "aud" (audience) claim is verified to ensure the token is intended for your
application, as identified by your client_id.

Example
{:ok, claims} = OpenIDConnect.verify(google_config, id_token)

# Common claims you might find in the response:
user_id = claims["sub"]         # Unique user identifier
email = claims["email"]         # User's email (if in scope)
name = claims["name"]           # User's name (if in scope)
picture = claims["picture"]     # URL to user's profile picture (if available)
issuer = claims["iss"]          # Identifies the token issuer
Ignoring "aud" claim verification:
google_config = %{client_id: "aaccaecd-fd29-46e7-be3e-58a99f355157", ...}

{:ok, claims} = OpenIDConnect.verify(google_config, id_token, ignore_claims: ["aud"])

audience = claims["aud"]        # b52bc817-2ab3-4aba-bd6e-286713d542f0
Security Warning
Always verify tokens before trusting their contents. Never use token data for
authentication purposes without verification, as tokens could be forged or tampered with.

        This module caches OIDC documents and their JWKs for a limited timeframe, which is min(@refresh_time, document.remaining_lifetime).



    


    
      
        fetch_document(uri)

      


        Fetches an OIDC document from cache, falling back to fetching from the remote resource

          @type t() :: %OpenIDConnect.Document{
  authorization_endpoint: term(),
  claims_supported: term(),
  end_session_endpoint: term(),
  expires_at: term(),
  jwks: term(),
  raw: term(),
  response_types_supported: term(),
  token_endpoint: term(),
  userinfo_endpoint: term()
}

This module caches OIDC documents and their JWKs for a limited timeframe, which is min(@refresh_time, document.remaining_lifetime).

          @spec fetch_document(String.t()) :: {:ok, t()} | {:error, term()}


      


Fetches an OIDC document from cache, falling back to fetching from the remote resource

  


        

      


  
OEBPS/dist/epub-4WIP524F.js
